§9 Geometrische Wahrscheinlichkeit

```
Bertrand1 := Module[{s, t, u, v, k, d, a, b},
    s = Sqrt[3]/2; t = 1/2;
    u = RandomReal[{−1, 1}]; v = RandomReal[{−Sqrt[1 − u^2], Sqrt[1 − u^2]}];
    k = −u/v; d = {u, v}.{u, v}/v;
    {a, b} = {x, y} /.NSolve[{y == k x + d, x^2 + y^2 == 1}, {x, y}];
    k1 = {Yellow, Disk[{0, 0}, 1]}; k2 = {Pink, Disk[{0, 0}, t]};
    k3 = {Yellow, Disk[{4, 0}, 1]}; k4 = {Pink, Disk[{4, 0}, t]};
    dr = Line[{{−s, −t}, {s, −t}, {0, 1}, {−s, −t}}];
    l1 = {Blue, Thickness[0.007], Line[{a, b}]}; l2 = Line[{{2.8, 0}, {5.1, 0}}]; l3 = Line[{{4, −1.2}, {4, 1.1}}];
    p1 = {Blue, PointSize[0.02], Point[{4 + u, v}]};
    t1 = Text["x", {5.3, 0}]; t2 = Text["y", {4, 1.3}]; t3 = Text["Ω", {4.9, 0.9}];
    Show[Graphics[{k1, k2, l1, dr, k3, k4, p1, l2, l3, t1, t2, t3}]]]

Bertrand2 := Module[{s, t, r, φ, u, v, k, d, a, b},
    s = Sqrt[3]/2; t = 1/2;
    r = RandomReal[]; φ = RandomReal[{0, 2 π}]; u = Cos[φ]; v = r Sin[φ];
    k = −u/v; d = {u, v}.{u, v}/v;
    {a, b} = {x, y} /.NSolve[{y == k x + d, x^2 + y^2 == 1}, {x, y}];
    k1 = {Yellow, Disk[{0, 0}, 1]}; k2 = {Pink, Disk[{0, 0}, t]};
    dr = Line[{{−s, −t}, {s, −t}, {0, 1}, {−s, −t}}];
    l1 = {Blue, Thickness[0.007], Line[{a, b}]}; l2 = Line[{{3.3, −1}, {4.6, −1}}];
    l3 = Line[{{3.5, −1.2}, {3.5, 1.1}}];
    r1 = {Yellow, Rectangle[{3.5, −1}, {4.5, 1}]}; r2 = {Pink, Rectangle[{3.5, −1}, {4, 1}]};
    p1 = {Blue, PointSize[0.022], Point[{3.5 + r, −1 + φ/π}]};
    t1 = Text["r", {4.8, −1}]; t2 = Text["φ", {3.5, 1.3}]; t3 = Text["Ω", {4.7, 0.8}];
    Show[Graphics[{k1, k2, l1, dr, r1, r2, p1, l2, l3, t1, t2, t3}]]]

ZerbrechenEinesStabes := Module[{x, y},
    x = RandomReal[]; y = RandomReal[];
    l1 = Line[{{0, 0}, {1, 0}}];
    p1 = {PointSize[0.03], Red, Point[{x, 0}]}; p2 = {PointSize[0.03], Blue, Point[{y, 0}]};
    p3 = {PointSize[0.03], Point[{0, 0}]}; p4 = {PointSize[0.03], Point[{1, 0}]};
    t1 = Text["0", {0, −0.1}]; t2 = Text["1", {1, −0.1}]; t3 = Text["x", {x, −0.1}]; t4 = Text["y", {y, −0.1}];
    Show[Graphics[{l1, p1, p2, p3, p4, t1, t2, t3, t4}], AspectRatio → 0.09]]

WartenAufBus := Module[{x, y},
    x = RandomReal[{0, 4}]; y = RandomReal[{0, 6}];
    l1 = {Red, Thickness[0.02], Line[{{0.02, 1}, {x, 1}}]}; l2 = {Blue, Thickness[0.02], Line[{{0.02, 2}, {y, 2}}]};
    t1 = Text["x", {x + 0.3, 1}, {−1, 0}]; t2 = Text["y", {y + 0.3, 2}, {−1, 0}];
    t3 = Text["    Zeit", {6.1, 0}, {−1, 0}];
    Show[Graphics[{l1, l2, t1, t2, t3}], AspectRatio → 0.2, Axes → True, Ticks → {Range[6], {}}]]
```

```
Dreieck := Module[{x, y, z, p, q},
    x = RandomReal[]; y = RandomReal[]; z = RandomReal[];
    {p, q} = Abs[{p, q} /.NSolve[{p^2 + q^2 == z^2, (x − p)^2 + q^2 == y^2}, {p, q}][[1]]];
    l1 = Line[{{−0.2, −0.2}, {1.2, −0.2}}]; l2 = Line[{{0, −0.2}, {0, 0.7}}];
    l3 = {Red, Thickness[0.015], Line[{{0, 0}, {x, 0}}]}; l4 = {Green, Thickness[0.015], Line[{{0, 0.3}, {y, 0.3}}]};
    l5 = {Blue, Thickness[0.015], Line[{{0, 0.6}, {z, 0.6}}]};
    t1 = Text["x", {x + 0.1, 0}, {−1, 0}]; t2 = Text["y", {y + 0.1, 0.3}, {−1, 0}];
    t3 = Text["z", {z + 0.1, 0.6}, {−1, 0}];
    t4 = Text["0", {−0.05, −0.35}, {−1, 0}]; t5 = Text["1", {0.9, −0.35}, {−1, 0}];
    ze = {{Red, Thickness[0.015], Line[{{2, 0}, {2 + x, 0}}]},
        {Green, Thickness[0.015], Line[{{2 + x, 0}, {2 + p, q}}]},
        {Blue, Thickness[0.015], Line[{{2 + p, q}, {2, 0}}]}};
    te = Text["nicht möglich", {2.5, 0}];
    eck = If[x + y > z && y + z > x && z + x > y, ze, te];
    p1 = {White, PointSize[0], Point[{−0.1, −0.1}]};
    p2 = {White, PointSize[0], Point[{3, 0.7}]};
    Show[Graphics[{l1, l2, l3, l4, l5, t1, t2, t3, t4, t5, eck, p1, p2}]]]

QuadratischeGleichung := Module[{p, q, x},
    p = RandomReal[]; q = RandomReal[];
    x = x /.NSolve[x^2 + 2 p x + q == 0, x];
    l1 = Line[{{−0.2, −0.2}, {1.2, −0.2}}]; l2 = Line[{{0, −0.2}, {0, 0.3}}];
    l3 = {Red, Thickness[0.02], Line[{{0.01, 0}, {p, 0}}]};
    l4 = {Green, Thickness[0.02], Line[{{0.01, 0.2}, {q, 0.2}}]};
    t1 = Text["p", {p + 0.1, 0}, {−1, 0}]; t2 = Text["q", {q + 0.1, 0.2}, {−1, 0}];
    t3 = Text["0", {−0.02, −0.3}, {−1, 0}]; t4 = Text["1", {1, −0.3}, {−1, 0}];
    t5 = If[Im[x[[1]]] == 0, Text[x, {2, 0}], Text["keine reellen Lösungen", {2, 0}]];
    Show[Graphics[{l1, l2, l3, l4, t1, t2, t3, t4, t5}]]]

Buffon := Module[{y, φ, a, b, c, d},
    y = RandomReal[]; φ = RandomReal[{0, π}];
    a = {0.2 + Cos[π/3]/2, 0.4+ Sin[π/6]/2}; b = {0.2 − Cos[π/3]/2, 0.4 − Sin[π/6]/2};
    c = {2.5 + Cos[φ]/3, y + Sin[φ]/3}; d = {2.5 − Cos[φ]/3, y − Sin[φ]/3};
    l11 = {Thickness[0.015], Line[{{0, 0}, {1, 0}}]}; l12 = {Thickness[0.015], Line[{{2, 0}, {3, 0}}]};
    l21 = {Thickness[0.015], Line[{{0, 1}, {1, 1}}]}; l22 = {Thickness[0.015], Line[{{2, 1}, {3, 1}}]};
    l3 = {Blue, Line[{{0.2, 0}, {0.2, 1}}]}; l4 = {Blue, Line[{{0.2, 0.4}, {0.5, 0.4}}]};
    l5 = {Thickness[0.015], RGBColor[1, 0, 1], Line[{a, b}]}; l6 = {Thickness[0.015], Red, Line[{c, d}]};
    pu = {PointSize[0.04], Point[{0.2, 0.4}]};
    t1 = Text["A", {0.05, 0.6}]; t2 = Text["φ", {0.45, 0.5}]; t3 = Text["y", {0.35, 0.2}];
    Show[Graphics[{l11, l12, l21, l22, l3, l4, l5, l6, pu, t1, t2, t3}], PlotRange → {−0.2, 1.2}]]
```

Manche Fragestellungen der elementaren Wahrscheinlichkeitsrechnung führen auf Probleme über Zufallsexperimente mit geometrischer Wahrscheinlichkeit. Wir klären in diesem Kapitel, was man unter einem derartigen Zufallsexperiment versteht und erläutern an Hand von Beispielen deren praktische Anwendung. Um die diesen Beispielen zu Grunde liegenden zufälligen Mechanismen besser verstehen zu können, wurden kleine Mathematica-Prozeduren entwickelt, mit denen sich diese Mechanismen simulieren lassen.

9.1 Geometrische Wahrscheinlichkeit

Das kontinuierliche Analogon zum Laplace-Experiment ist ein Zufallsexperiment mit geometrischer Wahrscheinlichkeit. Wir klären zuerst, was man unter einem derartigen Zufallsexperiment versteht und erläutern an einem Beispiel die mit der Gleichwahrscheinlichkeit der Realisierungen zusammenhängende Problematik.

9.1.1 Definition: Unter einem Zufallsexperiment mit geometrischer Wahrscheinlichkeit versteht man ein Zufallsexperiment, bei dem der Ereignisraum Ω eine Teilmenge des $\mathbb{R}^n$ ist, welche einen endlichen elementargeometrischen Inhalt (Länge, Flächeninhalt, Volumen, …) besitzt und die Wahrscheinlichkeit eines beliebigen Ereignisses $A \subseteq \Omega$ proportional zum Inhalt von A ist.

Beispiele für Zufallsexperimente mit geometrischer Wahrscheinlichkeit sind
- die zufällige Auswahl einer im Intervall $[0, 1]$ gleichverteilten Zahl;
- die zufällige Auswahl eines im Einheitsquadrat $[0, 1] \times [0, 1]$ gleichverteilten Punktes;
- das zufällige Eintreffen eines Signals im Zeitintervall $[0, T]$.

9.1.2 Bemerkung: Besitzt ein Zufallsexperiment mit geometrischer Wahrscheinlichkeit den Ereignisraum Ω, so ist das Wahrscheinlichkeitsmaß $\mathbb{P}$ auf Ω mit

$$\mathbb{P}[A] = \frac{\text{Inhalt von } A}{\text{Inhalt von } \Omega} = \frac{\text{Inhalt der günstigen Fälle}}{\text{Inhalt der möglichen Fälle}}$$

ein geeignetes Modell für den dieses Zufallsexperiment steuernden Zufall.

Achtung! Soll die Wahrscheinlichkeit $\mathbb{P}[A]$ eines Ereignisses A nach der Formel "Inhalt der günstigen Fälle durch Inhalt der möglichen Fälle" berechnet werden, so achte man bereits bei der Konstruktion des Ereignisraums Ω darauf, dass Ereignisse mit gleichem Inhalt auch tatsächlich gleich wahrscheinlich sind. Man überprüfe diese "Gleichwahrscheinlichkeit" stets durch ein Gedankenexperiment.

Mit einem berühmten Beispiel wollen wir die mit dieser Art von Zufallsexperimenten zusammenhängende Problematik verdeutlichen:

9.1.3 Beispiel (Paradoxon von BERTRAND): Wie groß ist die Wahrscheinlichkeit dafür, dass eine "zufällig" ausgewählte Sehne des Einheitskreises, welche nicht durch dessen Mittelpunkt geht, länger ist, als eine Seite des in diesen Kreis eingeschriebenen gleichseitigen Dreiecks?

▼

Lösung: Die folgende Zeichnung soll diese Fragestellung erläutern. Sie zeigt den (grau schattierten) Einheitskreis, ein in diesen Einheitskreis eingeschriebenes gleichseitiges Dreieck sowie eine zufällig ausgewählte Sehne des Einheitskreises, welche länger ist, als eine Seite dieses Dreiecks. Der Mittelpunkt dieser Sehne liegt damit innerhalb des (rot schattierten) Kreises mit Radius $1/2$.



1. Lösungsmöglichkeit: Jede nicht durch den Mittelpunkt des Einheitskreises gehende Sehne lässt sich eindeutig durch ihren Mittelpunkt $\{x, y\}$ beschreiben. Damit ist

$$\Omega = \{\{x, y\} \mid x, y \in \mathbb{R} \text{ mit } 0 < x^2 + y^2 < 1\}$$

ein möglicher Ereignisraum für dieses Zufallsexperiment. Das Ereignis "die gewählte Sehne ist länger als eine Seite des in diesen Einheitskreis eingeschriebenen gleichseitigen Dreiecks" entspricht damit der Menge

$$A = \{\{x, y\} \in \Omega \mid 0 < x^2 + y^2 < 1/4\}$$

und wir erhalten somit

$$\mathbb{P}[A] = \frac{\text{Flächeninhalt von } A}{\text{Flächeninhalt von } \Omega} = \frac{1}{4}$$

Mit dieser Wahl von Ω ist ein spezieller Zufallsmechanismus verbunden, mit dem eine Sehne des Einheitskreises ausgewählt wird. Dieser Zufallsmechanismus lässt sich mit dem Befehl Betrand1 simulieren. In der linken Zeichnung ist dabei die simulierte Sehne dargestellt, die rechte Zeichnung zeigt den dieser Sehne entsprechenden, im Ereignisraum Ω gleichverteilten Punkt:

Bertrand1



2. Lösungsmöglichkeit: Beschreibt man den Mittelpunkt einer derartigen Sehne nicht durch seine kartesischen Koordinaten $\{x, y\}$ sondern durch seine Polarkoordinaten $\{r, \varphi\}$, so ist

$$\Omega = \{\{r, \varphi\} \mid 0 < r < 1 \text{ mit } 0 \leq \varphi < 2\pi\}$$

ein möglicher Ereignisraum für dieses Zufallsexperiment. Das Ereignis "die gewählte Sehne ist länger als eine Seite des in diesen Einheitskreis eingeschriebenen gleichseitigen Dreiecks" entspricht dann der Menge

$$A = \{\{r, \varphi\} \in \Omega \mid 0 < r < 1/2\}$$

und wir erhalten somit

$$\mathbb{P}[A] = \frac{\text{Flächeninhalt von } A}{\text{Flächeninhalt von } \Omega} = \frac{1}{2}$$

Mit dieser Wahl von Ω ist ein ganz anderer Zufallsmechanismus verbunden, mit dem eine Sehne des Einheitskreises ausgewählt wird. Dieser Zufallsmechanismus lässt sich mit dem Befehl Betrand2 simulieren. In der linken Zeichnung ist dabei wieder die simulierte Sehne dargestellt, die rechte Zeichnung zeigt wieder den dieser Sehne entsprechenden, im Ereignisraum Ω gleichverteilten Punkt:

Bertrand2



Vergleicht man diese beiden Zufallsmechanismen, indem man damit längere Zeit experimentiert, so erkennt man, dass es sich dabei um verschiedene Mechanismen für die Auswahl einer Sehne des Einheitskreises handelt. Der Grund für die Verschiedenartigkeit dieser beiden Zufallsmechanismen liegt darin, dass kongruente Flächenelemente des ersten Ereignisraums keineswegs stets kongruenten Flächenelementen des zweiten Ereignisraums entsprechen. Beispielsweise treten beim zweiten Mechanismus häufiger Sehnen auf, deren Mittelpunkt in der Nähe des Ursprungs liegt. Wir haben es bei diesen beiden Zufallsmechanismen also um zwei verschiedene Verfahren der "zufälligen" Auswahl von Sehen zu tun.

Der Grund für das vermeintliche Paradoxon liegt somit letztlich in der hinterlistigen Fragestellung: Es wird nämlich nur von einer "zufällig" ausgewählten Sehne gesprochen ohne näher zu präzisieren, wie dieses "zufällige" Auswählen zu verstehen ist.

9.2 Beispiele

Wir behandeln nun Beispiele für Zufallsexperimente mit geometrischen Wahrscheinlichkeiten. Bei derartigen Beispielen empfiehlt es sich, den Ereignisraum Ω sowie das interessierende Ereignis A stets zeichnerisch darzustellen und zu überlegen, ob kongruente (infinitesimal kleine) Bereiche von Ω stets gleichwahrscheinlichen Ereignissen entsprechen. Oft muss dazu der in der Angabe nur oberflächlich erwähnte Begriff "zufällig" genauer interpretiert werden. Eine Simulation des der Fragestellung zu Grunde liegenden Zufallsexperiments ist oft hilfreich.

9.2.1 Beispiel: Ein Stab der Länge 1 wird zufällig in drei Teile zerlegt. Wie groß ist die Wahrscheinlichkeit dafür, dass man aus diesen drei Teilen ein Dreieck bilden kann?

▼

Lösung: Wir bezeichnen die beiden Bruchstellen mit x und y, und interpretieren den Ausdruck "zufällig" in der Weise, dass die beiden Bruchstellen im Intervall [0, 1] gleich verteilt sind (was eine sehr starke Vereinfachung der Realität darstellt). Der Zufallsmechanismus, der diese Art der Zerlegung des Stabes in drei Teile beschreibt, kann mit dem Befehl ZerbrechenEinesStabes simuliert werden:

ZerbrechenEinesStabes

Ein mathematisches Modell für diesen Zufallsmechanismus ist der Ereignisraum

$$\Omega = \{\{x,\ y\} \mid x,\ y \in [0,\ 1]\}$$

zusammen mit der geometrischen Wahrscheinlichkeit. Das Ereignis "aus den drei Teilstücken lässt sich ein Dreieck bilden" entspricht damit der Menge

$$A = \{\{x,\ y\} \in \Omega \mid \min[x,\ y] < 1/2 \bigwedge \max[x,\ y] - \min[x,\ y] < 1/2 \bigwedge 1 - \max[x,\ y] < 1/2\}$$

Wir veranschaulichen die Mengen Ω (gelb, zum Teil verdeckt) und A (blau)



und erkennen an Hand dieser Zeichnung, dass offenbar

$$\mathbb{P}[A] = \frac{\text{Flächeninhalt von } A}{\text{Flächeninhalt von } \Omega} = \frac{1}{4}$$

ist. Eine näherungsweise Ermittlung der Wahrscheinlichkeit dieses Ereignisses A durch Simulation liefert ein entsprechendes Resultat:

```
n = 10 000;
stichprobe = Table[RandomReal[{0, 1}, 2], {n}];
ereignis = Select[stichprobe, Min[#] < 1/2 && Max[#] − Min[#] < 1/2 && 1 − Max[#] < 1/2 &];
Length[ereignis]/n // N
Clear[n, stichprobe, ereignis]
```

```
0.2546
```

9.2.2 Beispiel: An einer Omnibushaltestelle trifft alle 4 Minuten ein Bus der Linie I und alle 6 Minuten ein Bus der Line II ein. Man bestimme die Wahrscheinlichkeit dafür, dass

- der nächste Bus ein Bus der Linie I ist;
- innerhalb der nächsten zwei Minuten ein Bus eintrifft.

▼

Lösung: Zunächst ist unklar, wieso bei dieser Fragestellung der Zufall überhaupt eine Rolle spielt. Interpretiert man aber die Fragestellung so, dass ab einem zufällig gewählten Zeitpunkt die Wartezeit x auf den nächsten Bus der Linie I im Intervall $[0, 4]$ und die Wartezeit y auf den nächsten Bus der Linie II im Intervall $[0, 6]$ gleichverteilt ist und die Busse unabhängig voneinander (also nicht im Takt) verkehren, so liegt dieser Fragestellung ein bestimmter Zufallsmechanismus zu Grunde, der mit dem Befehl WartenAufBus simuliert werden kann:

```
WartenAufBus
```



Ein mathematisches Modell für diesen Zufallsmechanismus ist der Ereignisraum

$$\Omega = \{\{x, y\} \mid 0 \le x \le 4 \text{ und } 0 \le y \le 6\}$$

zusammen mit der geometrischen Wahrscheinlichkeit. Die beiden Ereignisse "der nächste Bus ist ein Bus der Linie I" und "innerhalb der nächsten zwei Minuten trifft ein Bus ein" entsprechen damit den Mengen

$$A = \{\{x, y\} \in \Omega \mid x < y\} \quad \text{und} \quad B = \{\{x, y\} \in \Omega \mid \min[x, y] < 2\}$$

Wir veranschaulichen die Mengen Ω (gelb, zum Teil verdeckt), A (blau) und B (rot)



und erkennen an Hand dieser Zeichnung, dass offenbar

$$\mathbb{P}[A] = \frac{\text{Flächeninhalt von } A}{\text{Flächeninhalt von } \Omega} = \frac{2}{3} \quad \text{und} \quad \mathbb{P}[B] = \frac{\text{Flächeninhalt von } B}{\text{Flächeninhalt von } \Omega} = \frac{2}{3}$$

gilt. Eine näherungsweise Ermittlung der Wahrscheinlichkeiten dieser beiden Ereignisse A und B durch Simulation liefert ein entsprechendes Resultat:

```
n = 10 000;
stichprobe = Table[{RandomReal[{0, 4}], RandomReal[{0, 6}]}, {n}];
ereignisA = Select[stichprobe, #[[1]] < #[[2]] &];
ereignisB = Select[stichprobe, Min[#] < 2 &];
Length[ereignisA] / n // N
Length[ereignisB] / n // N
Clear[n, stichprobe, ereignisA, ereignisB]
```

```
0.662

0.6643
```

9.2.3 Beispiel: Wir betrachten drei Stäbe mit im Intervall [0, 1] gleichverteilten Längen x, y und z. Wie groß ist die Wahrscheinlichkeit dafür, dass sich aus diesen drei Stäben ein Dreieck bilden lässt?

▼

Lösung: Der Zufallsmechanismus, mit dem die Längen x, y, z der drei Stäbe ausgewählt werden, um daraus ein Dreieck zu bilden, ist offensichtlich. Dieser Mechanismus lässt sich mit dem Befehl Dreieck simulieren (falls ein Dreieck gebildet werden kann, so wird dieses Dreieck gezeichnet; falls kein Dreieck gebildet werden kann, so erscheint der Text "nicht möglich"):

```
Dreieck
```



Ein mathematisches Modell für diesen Zufallsmechanismus ist der Ereignisraum

$$\Omega = \{\{x,\ y,\ z\} \mid x,\ y,\ z \in [0,\ 1]\ \}$$

zusammen mit der geometrischen Wahrscheinlichkeit. Das Ereignis "aus den drei Stäben lässt sich ein Dreieck bilden" entspricht damit der Menge

$$A = \{\{x,\ y,\ z\} \in \Omega \mid x + y > z \text{ und } y + z > x \text{ und } z + x > y\}$$

Wir veranschaulichen diese beiden Mengen wieder zeichnerisch (die Menge Ω entspricht einem achsenparallelen Würfel mit der Kantenlänge 1; die Menge A erhält man, indem man von diesem Würfel Ω drei Pyramiden mit Spitzen in den Punkten {1, 0, 0}, {0, 1, 0} und {0, 0, 1} abschneidet)



und erkennen an Hand dieser Zeichnung, dass offenbar

$$\mathbb{P}[A] = \frac{\text{Volumen von } A}{\text{Volumen von } \Omega} = 1 - \frac{1}{2}\,\frac{1}{3} - \frac{1}{2}\,\frac{1}{3} - \frac{1}{2}\,\frac{1}{3} = \frac{1}{2}$$

ist. Eine näherungsweise Ermittlung der Wahrscheinlichkeit dieses Ereignisses A durch Simulation liefert wieder ein entsprechendes Resultat:

```
n = 10 000;
stichprobe = Table[RandomReal[{0, 1}, 3], {n}];
ereignis = Select[stichprobe, #[[1]] + #[[2]] > #[[3]] && #[[2]] + #[[3]] > #[[1]] && #[[3]] + #[[1]] > #[[2]] &];
Length[ereignis] / n // N
Clear[n, stichprobe, ereignis]
```

```
0.493
```

9.2.4 Beispiel: Die Zahlen p und q werden zufällig aus dem Intervall [0, 1] ausgewählt. Wie groß ist die Wahrscheinlichkeit dafür, dass die Lösungen der quadratischen Gleichung $x^2 + 2\,p\,x + q = 0$ reell sind?

▼

Lösung: Der Zufallsmechanismus, mit dem die beiden Zahlen p und q aus dem Intervall [0, 1] ausgewählt werden, ist offensichtlich. Er lässt sich mit dem Befehl QuadratischeGleichung simulieren (falls die Lösungen der quadratischen Gleichung $x^2 + 2\,p\,x + q = 0$ reell sind, so werden diese Lösungen in Form einer Liste ausgegeben; falls die Lösungen dieser Gleichung nicht reell sind, so erscheint der Text "keine reellen Lösungen")

```
QuadratischeGleichung
```



keine reellen Lösungen

und durch den Ereignisraum

$$\Omega = \{\{p, q\} \mid p, q \in [0, 1]\}$$

zusammen mit der geometrischen Wahrscheinlichkeit beschreiben. Da die Lösungen der quadratischen Gleichung $x^2 + 2\,p\,x + q = 0$ bekanntlich genau dann reell sind, wenn die Beziehung $p^2 \geq q$ gilt, entspricht das uns interessierende Ereignis "die Lösungen der quadratischen Gleichung $x^2 + 2\,p\,x + q = 0$ sind reell" der Menge

$$A = \{\{p, q\} \in \Omega \mid p^2 \geq q\}$$

Wir veranschaulichen wieder die Mengen Ω (gelb, zum Teil verdeckt) und A (blau)



und erkennen an Hand dieser Zeichnung, dass offensichtlich

$$\mathbb{P}[A] = \frac{\text{Flächeninhalt von } A}{\text{Flächeninhalt von } \Omega} = \int_0^1 p^2 \, dp = \frac{1}{3}$$

gilt. Eine näherungsweise Ermittlung der Wahrscheinlichkeit dieses Ereignisses A durch Simulation liefert wieder ein entsprechendes Resultat:

```
n = 10 000;
stichprobe = Table[RandomReal[{0, 1}, 2], {n}];
ereignis = Select[stichprobe, #[[1]]^2 ≥ #[[2]] &];
Length[ereignis] / n // N
Clear[n, stichprobe, ereignis]
```

```
0.3407
```

9.2.5 Beispiel (BUFFON'sches Nadelproblem): Wir ziehen in einer Ebene waagrechte, parallele Geraden im Abstand A und lassen auf diese Ebene zufällig Nadeln der Länge L fallen. Der Einfachheit halber nehmen wir an, dass $L < A$ ist, jede Nadel also höchstens eine dieser Gerade schneiden kann. Wie groß ist die Wahrscheinlichkeit dafür, dass eine zufällig auf diese Ebene fallende Nadel eine Gerade schneidet?

▼

Lösung: Die Lage der Nadel auf der Ebene ist durch ihren Mittelpunkt $\{x, y\}$ und durch den Winkel φ, den sie mit den Geraden einschließt, vollständig bestimmt. Die x-Koordinate des Mittelpunktes der Nadel hat dabei keinen Einfluss darauf, ob die Nadel eine Gerade schneidet. Wir können daher aus Symmetriegründen annehmen, dass der Mittelpunkt der Nadel auf einer Strecke zwischen zwei ausgewählten benachbarten parallelen Geraden liegt.

Es liegt nahe, den Zufallsmechanismus, mit dem eine Nadel auf die Ebene fällt, in der Weise zu interpretieren, dass die y-Koordinate des Mittelpunktes im Intervall $[0, A]$ gleichverteilt ist und (unabhängig davon) der Winkel φ, den die Nadel mit den Geraden einschließt, im Intervall $[0, \pi]$ gleichverteilt ist. Dieser Zufallsmechanismus lässt sich mit dem Befehl Buffon simulieren:

```
Buffon
```



Ein mathematisches Modell für diesen Zufallsmechanismus ist der Ereignisraum

$$\Omega = \{\{\varphi, y\} \mid 0 \le \varphi \le \pi \text{ und } 0 \le y \le A\}$$

zusammen mit der geometrischen Wahrscheinlichkeit. Das Ereignis "die Nadel schneidet eine Gerade" entspricht damit der Menge

$$S = \{\{\varphi, y\} \in \Omega \mid y \le \frac{L}{2} \operatorname{Sin}[\varphi] \text{ oder } y \ge A - \frac{L}{2} \operatorname{Sin}[\varphi]\}$$

Wir veranschaulichen die Mengen Ω (gelb, zum Teil verdeckt) und S (blau) wieder zeichnerisch



und erkennen an Hand dieser Zeichnung, dass offenbar

$$\mathbb{P}[S] = \frac{\text{Flächeninhalt von } S}{\text{Flächeninhalt von } \Omega} = \frac{2}{A\,\pi} \int_0^{\pi} \frac{L}{2} \operatorname{Sin}[\varphi]\, d\varphi = \frac{L}{A\,\pi}$$

ist. Eine näherungsweise Ermittlung der Wahrscheinlichkeit dieses Ereignisses S durch Simulation liefert ebenfalls ein entsprechendes Resultat:

```
L = 2; A = 3; n = 10 000;
stichprobe = Table[{RandomReal[{0, 2 π}], RandomReal[{0, A}]}, {n}];
ereignis = Select[stichprobe, #[[2]] ≤ L Sin[#[[1]]] / 2 || #[[2]] ≥ A − L Sin[#[[1]]] / 2 &];
Length[ereignis] / n // N
Clear[L, A, n, stichprobe, ereignis]
```

```
0.2026
```